

---

Subject: Kenya DHS 2014; which sample weight to use?

Posted by [gatcho](#) on Thu, 13 Apr 2017 13:23:22 GMT

[View Forum Message](#) <> [Reply to Message](#)

---

Hi all,

I am working on the determinants of the health insurance ownership among the household head in Kenya with the DHS dataset of Kenya 2014 : KEHR70FL ; KEMR70FL ; KEIR70FL.

I am a beginner in Stata and DHS data user

The unit of analysis is the household head.

I have merged these three datasets (KEHR70FL ; KEMR70FL ; KEIR70FL) because the question about health insurance ownership exists only in the man and woman database.

For the variable v481 (covered by health insurance), there is 14636 household head with 3576 missing values.

My question :

1- What sample weight can I use to have representative results at the national level?

Thanks.

Modeste

---

---

Subject: Re: Kenya DHS 2014; which sample weight to use?

Posted by [Bridgette-DHS](#) on Thu, 13 Apr 2017 15:18:14 GMT

[View Forum Message](#) <> [Reply to Message](#)

---

Following is a response from Senior DHS Stata Specialist, Tom Pullum:

I checked what you did with the Stata lines given below. I agree with what you got. I recommend that you construct a new weight that is v005 for the women and mv005 for the men. I call it v005\_mv005 but you could call it whatever you want.

```
use e:\DHS\DHS_data\PR_files\KEPR70FL.dta, clear
keep hv001 hv002 hvidx hv005 hv101 hv104 hv105
keep if hv101==1
sort hv001 hv002 hvidx
save e:\DHS\DHS_data\scratch\PRtemp.dta, replace
```

```
use e:\DHS\DHS_data\IR_files\KEIR70FL.dta, clear
keep v001 v002 v003 v005 v481
rename v001 hv001
rename v002 hv002
rename v003 hvidx
tab v481,m
```

```
gen sex=2
save e:\DHS\DHS_data\scratch\IRtemp.dta, replace
```

```
use e:\DHS\DHS_data\MR_files\KEMR70FL.dta, clear
```

```
keep mv001 mv002 mv003 mv005 mv481
rename mv001 hv001
rename mv002 hv002
rename mv003 hvidx
tab mv481,m
```

```
rename mv481 v481
gen sex=1
save e:\DHS\DHS_data\scratch\MRtemp.dta, replace
append using e:\DHS\DHS_data\scratch\IRtemp.dta
sort hv001 hv002 hvidx
```

```
merge hv001 hv002 hvidx using e:\DHS\DHS_data\scratch\PRtemp.dta
```

```
tab _merge
keep if _merge==3
drop _merge
tab v481 sex,m
```

```
gen v005_mv005=.
replace v005_mv005==mv005 if sex==1
replace v005_mv005==v005 if sex==2
```

---

Subject: Re: Kenya DHS 2014; which sample weight to use?

Posted by [gatcho](#) on Fri, 14 Apr 2017 09:16:52 GMT

[View Forum Message](#) <> [Reply to Message](#)

---

Hi,  
Thank you for your suggestion.

Just at the end of your code, it's work with one equal.

```
replace v005_mv005=mv005 if sex==1.
```

```
replace v005_mv005=v005 if sex==2
```

Thanks for the prompt response.

---

Subject: Re: Kenya DHS 2014; which sample weight to use?

Posted by [Bridgette-DHS](#) on Fri, 14 Apr 2017 11:47:17 GMT

[View Forum Message](#) <> [Reply to Message](#)

---

From Senior DHS Stata Specialist, Tom Pullum:

Thanks for fixing that! Tom

---